
Pesquisa em Memória Primária*

Última alteração: 26 de Março de 2004

*Transparências elaboradas por Fabiano C. Botelho e Nivio Ziviani

Pesquisa em Memória Primária

- Introdução - Conceitos Básicos
- Pesquisa Seqüencial
- Pesquisa Binária
- Árvores de Pesquisa
 - Árvores Binárias de Pesquisa sem Balanceamento
 - Árvores Binárias de Pesquisa com Balanceamento
 - * Árvores SBB
 - * Transformações para Manutenção da Propriedade SBB
- Pesquisa Digital
 - Trie
 - Patricia
- Transformação de Chave (*Hashing*)
 - Funções de Transformação
 - Listas Encadeadas
 - Endereçamento Aberto
 - *Hashing* Perfeito

Introdução - Conceitos Básicos

- estudo de como recuperar informação a partir de uma grande massa de informação previamente armazenada.
- A informação é dividida em **registros**.
- Cada registro possui uma chave para ser usada na pesquisa.
- **Objetivo da pesquisa:**
Encontrar uma ou mais ocorrências de registros com chaves iguais à chave de pesquisa.
- **Pesquisa com sucesso X Pesquisa sem sucesso.**

Introdução - Conceitos Básicos

Tabelas

- Conjunto de registros ou arquivos \Rightarrow
TABELAS
- **Tabela:**
associada a entidades de vida curta, criadas na memória interna durante a execução de um programa.
- **Arquivo:**
geralmente associado a entidades de vida mais longa, armazenadas em memória externa.
- **Distinção não é rígida:**
tabela: arquivo de índices
arquivo: tabela de valores de funções.

Escolha do Método de Pesquisa mais Adequado a uma Determinada Aplicação

- **Depende principalmente:**
 1. Quantidade dos dados envolvidos.
 2. Arquivo estar sujeito a inserções e retiradas freqüentes.

se conteúdo do arquivo é estável é importante minimizar o tempo de pesquisa, sem preocupação com o tempo necessário para estruturar o arquivo

Algoritmos de Pesquisa \Rightarrow Tipos Abstratos de Dados

- É importante considerar os algoritmos de pesquisa como **tipos abstratos de dados**, com um conjunto de operações associado a uma estrutura de dados, de tal forma que haja uma independência de implementação para as operações.
- **Operações mais comuns:**
 1. Inicializar a estrutura de dados.
 2. Pesquisar um ou mais registros com determinada chave.
 3. Inserir um novo registro.
 4. Retirar um registro específico.
 5. Ordenar um arquivo para obter todos os registros em ordem de acordo com a chave.
 6. Ajuntar dois arquivos para formar um arquivo maior.

Dicionário

- Nome comumente utilizado para descrever uma estrutura de dados para pesquisa.
- **Dicionário** é um **tipo abstrato de dados** com as operações:
 1. Inicializa
 2. Pesquisa
 3. Insere
 4. Retira
- Analogia com um dicionário da língua portuguesa:
 - Chaves \iff palavras
 - Registros \iff entradas associadas com cada palavra:
 - * pronúncia
 - * definição
 - * sinônimos
 - * outras informações

Pesquisa Seqüencial

- **Método de pesquisa mais simples:** a partir do primeiro registro, pesquise seqüencialmente até encontrar a chave procurada; então pare.
- Armazenamento de um conjunto de registros por meio do tipo estruturado arranjo:

```
const Maxn = 1000;
```

```
type Registro = record
```

```
    Chave: TipoChave;
```

```
    {outros componentes}
```

```
end;
```

```
Indice = 0..Maxn;
```

```
Tabela = record
```

```
    Item: array [Indice] of Registro;
```

```
    n : Indice;
```

```
end;
```

Pesquisa Seqüencial

Implementação para as operações Inicializa, Pesquisa e Insere:

```
procedure Inicializa (var T: Tabela);
```

```
begin T.n := 0; end;
```

```
function Pesquisa (x:TipoChave;var T:Tabela):Indice;
```

```
var i: integer;
```

```
begin
```

```
    T.Item[0].Chave := x; i := T.n + 1;
```

```
    repeat
```

```
        i := i - 1;
```

```
    until T.Item[i].Chave = x;
```

```
    Pesquisa := i;
```

```
end;
```

```
procedure Insere (Reg: Registro; var T: Tabela);
```

```
begin
```

```
    if T.n = Maxn
```

```
    then writeln('Erro: tabela cheia')
```

```
    else begin
```

```
        T.n := T.n + 1; T.Item[T.n] := Reg;
```

```
    end;
```

```
end;
```

Pesquisa Seqüencial

- Pesquisa retorna o índice do registro que contém a chave x ;
- Caso não esteja presente, o valor retornado é zero.
- A implementação não suporta mais de um registro com uma mesma chave.
- Para aplicações com esta característica é necessário incluir um argumento a mais na função Pesquisa para conter o índice a partir do qual se quer pesquisar.

Pesquisa Seqüencial

- Utilização de um registro **sentinela** na posição zero do **array**:
 1. Garante que a pesquisa sempre termina: se o índice retornado por Pesquisa for zero, a pesquisa foi sem sucesso.
 2. Não é necessário testar se $i > 0$, devido a isto:
 - o anel interno da função Pesquisa é extremamente simples: o índice i é decrementado e a chave de pesquisa é comparada com a chave que está no registro.
 - isto faz com que esta técnica seja conhecida como **pesquisa seqüencial rápida**.

Pesquisa Seqüencial

Análise

- Pesquisa com sucesso:

$$\text{melhor caso} : C(n) = 1$$

$$\text{pior caso} : C(n) = n$$

$$\text{caso médio} : C(n) = (n + 1)/2$$

- Pesquisa sem sucesso:

$$C'(n) = n + 1.$$

- O algoritmo de pesquisa seqüencial é a **melhor escolha** para o problema de pesquisa em tabelas com até **25 registros**.

Pesquisa Binária

- **Pesquisa em tabela pode ser mais eficiente \Rightarrow Se registros forem mantidos em ordem**
- Para saber se uma chave está presente na tabela
 1. Compare a chave com o registro que está na posição do meio da tabela.
 2. **Se** a chave é menor **então** o registro procurado está na primeira metade da tabela
 3. **Se** a chave é maior **então** o registro procurado está na segunda metade da tabela.
 4. Repita o processo até que a chave seja encontrada, ou fique apenas um registro cuja chave é diferente da procurada, significando uma pesquisa sem sucesso.

Exemplo de Pesquisa Binária para a Chave G

	1	2	3	4	5	6	7	8
Chaves iniciais:	<i>A</i>	<i>B</i>	<i>C</i>	<i>D</i>	<i>E</i>	<i>F</i>	<i>G</i>	<i>H</i>
	<i>A</i>	<i>B</i>	<i>C</i>	<i>D</i>	<i>E</i>	<i>F</i>	<i>G</i>	<i>H</i>
					<i>E</i>	<i>F</i>	<i>G</i>	<i>H</i>
							<i>G</i>	<i>H</i>

Algoritmo de Pesquisa binária

```
function Binaria (x: TipoChave; var T: Tabela): Indice;  
var i, Esq, Dir: Indice;  
begin  
    if T.n = 0  
    then Binaria := 0  
    else begin  
        Esq := 1; Dir := T.n;  
        repeat  
            i := (Esq + Dir) div 2;  
            if x > T.Item[i].Chave  
            then Esq := i+1  
            else Dir := i-1;  
        until (x = T.Item[i].Chave) or (Esq > Dir);  
        if x = T.Item[i].Chave  
        then Binaria := i  
        else Binaria := 0;  
    end;  
end;
```

Pesquisa Binária

Análise


- A cada iteração do algoritmo, o tamanho da tabela é dividido ao meio.
- **Logo:** o número de vezes que o tamanho da tabela é dividido ao meio é cerca de $\log n$.
- **Ressalva:** o custo para manter a tabela ordenada é alto:
a cada inserção na posição p da tabela implica no deslocamento dos registros a partir da posição p para as posições seguintes.
- Conseqüentemente, a pesquisa binária não deve ser usada em aplicações muito dinâmicas.

Árvores de Pesquisa

- A árvore de pesquisa é uma estrutura de dados muito eficiente para armazenar informação.
- Particularmente adequada quando existe necessidade de considerar todos ou alguma combinação de:
 1. Acesso direto e seqüencial eficientes.
 2. Facilidade de inserção e retirada de registros.
 3. Boa taxa de utilização de memória.
 4. Utilização de memória primária e secundária.

Árvores Binárias de Pesquisa sem Balanceamento

- Para qualquer nó que contenha um registro




Temos a relação invariante



1. Todos os registros com chaves menores estão na subárvore à esquerda.
2. Todos os registros com chaves maiores estão na subárvore à direita.

Árvores Binárias de Pesquisa sem Balanceamento

Exemplo



- O **nível** do nó raiz é 0.
- Se um nó está no nível i então a raiz de suas subárvores estão no nível $i + 1$.
- A **altura** de um nó é o comprimento do caminho mais longo deste nó até um nó folha.
- A altura de uma árvore é a altura do nó raiz.

Implementação do Tipo Abstrato de Dados Dicionário usando a Estrutura de Dados Árvore Binária de Pesquisa

Estrutura de dados:

```
type TipoChave = integer;  
  Registro = record  
    Chave: TipoChave;  
    { outros componentes }  
  end;  
  Apontador = ^No;  
  No = record  
    Reg: Registro;  
    Esq, Dir: Apontador;  
  end;  
  TipoDicionario = Apontador;
```

Procedimento para Pesquisar na Árvore

Para encontrar um registro com uma chave x :

- Compare-a com a chave que está na raiz.
- Se x é menor, vá para a subárvore esquerda.
- Se x é maior, vá para a subárvore direita.
- Repita o processo recursivamente, até que a chave procurada seja encontrada ou um nó folha é atingido.
- Se a pesquisa tiver sucesso então o conteúdo do registro retorna no próprio registro x .

```
procedure Pesquisa (var x: Registro; var p: Apontador);  
begin  
  if p = nil  
  then writeln( 'Erro:Registro nao esta presente na  
                arvore ' )  
  else if x.Chave < p^.Reg.Chave  
    then Pesquisa(x, p^.Esq)  
    else if x.Chave > p^.Reg.Chave  
      then Pesquisa(x, p^.Dir)  
      else x := p^.Reg;  
end;
```

Procedimento para Inserir na Árvore

- Atingir um apontador nulo em um processo de pesquisa significa uma pesquisa sem sucesso.
- O apontador nulo atingido é o ponto de inserção.

```
procedure Insere (x: Registro; var p: Apontador);  
begin  
  if p = nil  
  then begin  
    new(p); p^.Reg := x;  
    p^.Esq := nil; p^.Dir := nil;  
  end  
  else if x.Chave < p^.Reg.Chave  
  then Insere(x, p^.Esq)  
  else if x.Chave > p^.Reg.Chave  
  then Insere(x, p^.Dir)  
  else writeln('Erro: Registro ja existe  
                na arvore')  
end;
```

Procedimentos para Inicializar e Criar a Árvore

```
procedure Inicializa (var Dicionario: TipoDicionario);  
begin  
    Dicionario := nil;  
end;
```


```
program CriaArvore;  
type TipoChave = integer;  
{— Entra aqui a definição dos tipos mostrados na transp. 19 —}  
var Dicionario: TipoDicionario;  
    x: Registro;  
{— Entram aqui os procedimentos Insere e Inicializa —}  
begin  
    Inicializa (Dicionario);  
read (x.Chave);  
while x.Chave > 0 do  
    begin  
        Insere (x, Dicionario);  
        read (x.Chave);  
    end;  
end.
```

Procedimento para Retirar x da Árvore

- **Alguns comentários:**

1. A retirada de um registro não é tão simples quanto a inserção.
2. Se o nó que contém o registro a ser retirado possui no máximo um descendente \Rightarrow a operação é simples.
3. No caso do nó conter dois descendentes o registro a ser retirado deve ser primeiro:
 - substituído pelo registro mais à direita na subárvore esquerda;
 - ou pelo registro mais à esquerda na subárvore direita.

Exemplo da Retirada de um Registro da Árvore



Assim: para retirar o registro com chave 5 na árvore basta trocá-lo pelo registro com chave 4 ou pelo registro com chave 6, e então retirar o nó que recebeu o registro com chave 5.

Procedimento para Retirar x da Árvore

```
procedure Retira (x: Registro; var p: Apontador);  
var Aux: Apontador;  
    procedure Antecessor (q: Apontador; var r: Apontador);  
    begin  
        if r^.Dir <> nil  
        then Antecessor(q, r^.Dir)  
        else begin  
            q^.Reg := r^.Reg;  
            q := r; r := r^.Esq;  
            dispose(q)  
        end;  
    end;
```

Procedimento para Retirar x da Árvore




```

begin {Retira}
  if p = nil
  then writeln('Erro: Registro nao esta na arvore')
  else if x.Chave < p^.Reg.Chave
    then Retira(x, p^.Esq)
    else if x.Chave > p^.Reg.Chave
      then Retira(x, p^.Dir)
      else if p^.Dir = nil
        then begin
          Aux := p; p := p^.Esq;
          dispose(Aux);
        end
      else if p^.Esq = nil
        then begin
          Aux:=p; p:=p^.Dir;
          dispose(Aux);
        end
      else Antecessor(p, p^.Esq);
end; {Retira}

```

- **Obs.:** proc. recursivo Antecessor só é ativado quando o nó que contém registro a ser retirado possui 2 descendentes. Solução usada por Wirth, 1976, p.211.

Outro Exemplo de Retirada de Nó




Caminhamento Central

- Após construída a árvore, pode ser necessário percorrer todos os registros que compõem a tabela ou arquivo.
- Existe mais de uma ordem de **caminhamento** em árvores, mas a mais útil é a chamada ordem de **caminhamento central**.
- O caminhamento central é mais bem expresso em termos recursivos:
 1. caminha na subárvore esquerda na ordem central;
 2. visita a raiz;
 3. caminha na subárvore direita na ordem central.
- Uma característica importante do caminhamento central é que os nós são visitados de forma ordenada.

Caminhamento Central

- Percorrer a árvore:



usando caminhamento central recupera as chaves na ordem 1, 2, 3, 4, 5, 6 e 7.

- O procedimento Central é mostrado abaixo:

```
procedure Central (p: Apontador);  
begin  
  if p <> nil  
  then begin  
    Central(p^.Esq); writeln(p^.Reg.Chave);  
    Central(p^.Dir);  
  end;  
end;
```

Análise

- O número de comparações em uma pesquisa com sucesso:

melhor caso : $C(n) = O(1)$

pior caso : $C(n) = O(n)$

caso médio : $C(n) = O(\log n)$


- O tempo de execução dos algoritmos para árvores binárias de pesquisa dependem muito do formato das árvores.

Análise

1. Para obter o pior caso basta que as chaves sejam inseridas em ordem crescente ou decrescente. Neste caso a árvore resultante é uma lista linear, cujo número médio de comparações é $(n + 1)/2$.
2. Para uma **árvore de pesquisa randômica** o número esperado de comparações para recuperar um registro qualquer é cerca de $1,39 \log n$, apenas 39% pior que a árvore completamente balanceada.
 - Uma árvore A com n chaves possui $n + 1$ nós externos e estas n chaves dividem todos os valores possíveis em $n + 1$ intervalos. Uma inserção em A é considerada *randômica* se ela tem probabilidade igual de acontecer em qualquer um dos $n + 1$ intervalos.
 - Uma *árvore de pesquisa randômica* com n chaves é uma árvore construída através de n inserções randômicas sucessivas em uma árvore inicialmente vazia.

Árvores Binárias de Pesquisa com Balanceamento

- Árvore completamente balanceada \Rightarrow nós externos aparecem em no máximo dois níveis adjacentes.
- Minimiza tempo médio de pesquisa para uma distribuição uniforme das chaves, onde cada chave é igualmente provável de ser usada em uma pesquisa.
- Contudo, custo para manter a árvore completamente balanceada após cada inserção é muito alto.
- Para inserir a chave 1 na árvore do exemplo à esquerda e obter a árvore à direita do mesmo exemplo é necessário movimentar todos os nós da árvore original.
- **Exemplo:**



Uma Forma de Contornar este Problema


- Procurar solução intermediária que possa manter árvore “quase-balanceada”, em vez de tentar manter a árvore completamente balanceada.
- **Objetivo:** Procurar obter bons tempos de pesquisa, próximos do tempo ótimo da árvore completamente balanceada, mas sem pagar muito para inserir ou retirar da árvore.
- **Heurísticas:** existem várias heurísticas baseadas no princípio acima.
- Gonnet e Baeza-Yates (1991) apresentam algoritmos que utilizam vários critérios de balanceamento para árvores de pesquisa, tais como restrições impostas:
 - na diferença das alturas de subárvores de cada nó da árvore,
 - na redução do **comprimento do caminho interno**
 - ou que todos os nós externos apareçam no mesmo nível.

Uma Forma de Contornar este Problema

- **Comprimento do caminho interno:**
corresponde à soma dos comprimentos dos caminhos entre a raiz e cada um dos nós internos da árvore.
- Por exemplo, o comprimento do caminho interno da árvore à esquerda na figura da transparência anterior é
 $8 = (0 + 1 + 1 + 2 + 2 + 2)$.

Árvores SBB

- Árvores B \Rightarrow estrutura para memória secundária. (Bayer R. e McCreight E.M., 1972)
- **Árvore 2-3** \Rightarrow caso especial da árvore B.
- Cada nó tem duas ou três subárvores.
- Mais apropriada para memória primária.
- **Exemplo: Uma árvore 2-3 e a árvore B binária correspondente**(Bayer, R. 1971)




Árvores SBB

- **Árvore 2-3** \Rightarrow **árvore B binária** (assimetria inerente)
 1. Apontadores à esquerda apontam para um nó no nível abaixo.
 2. Apontadores à direita podem ser verticais ou horizontais.



Eliminação da assimetria nas árvores B binárias \Rightarrow árvores B binárias simétricas (*Symmetric Binary B-trees* – SBB)

- **Árvore SBB** é uma árvore binária com 2 tipos de apontadores: verticais e horizontais, tal que:
 1. todos os caminhos da raiz até cada nó externo possuem o mesmo número de apontadores verticais, e
 2. não podem existir dois apontadores horizontais sucessivos.



Transformações para Manutenção da Propriedade SBB

- O algoritmo para árvores SBB usa transformações locais no caminho de inserção ou retirada para preservar o balanceamento.
- A chave a ser inserida ou retirada é sempre inserida ou retirada após o apontador vertical mais baixo na árvore.
- Dependendo da situação anterior à inserção ou retirada, podem aparecer dois apontadores horizontais sucessivos
- **Neste caso:** é necessário realizar uma transformação.
- **Transformações Propostas por Bayer R. 1972**



Estrutura de Dados Árvore SBB para Implementar o Tipo Abstrato de Dados Dicionário

```
type TipoChave = integer;  
      Registro   = record  
                  Chave: TipoChave  
                  { outros componentes }  
                end;  
      Inclinacao = (Vertical , Horizontal);  
      Apontador  = ^No;  
      No = record  
          Reg      : Registro;  
          Esq, Dir  : Apontador;  
          BitE , BitD: Inclinacao  
        end;  
      TipoDicionario = Apontador;
```

Procedimentos Auxiliares para Árvores SBB

```
procedure EE (var Ap: Apontador);  
var Ap1: Apontador;  
begin  
    Ap1 := Ap^.Esq; Ap^.Esq := Ap1^.Dir; Ap1^.Dir := Ap;  
    Ap1^.BitE := Vertical; Ap^.BitE := Vertical;  
    Ap := Ap1;  
end;
```

```
procedure ED (var Ap: Apontador);  
var Ap1, Ap2: Apontador;  
begin  
    Ap1 := Ap^.Esq; Ap2 := Ap1^.Dir;  
    Ap1^.BitD := Vertical; Ap^.BitE := Vertical;  
    Ap1^.Dir := Ap2^.Esq; Ap2^.Esq := Ap1;  
    Ap^.Esq := Ap2^.Dir; Ap2^.Dir := Ap; Ap := Ap2;  
end;
```

Procedimentos Auxiliares para Árvores SBB

```
procedure DD (var Ap: Apontador);  
var Ap1: Apontador;  
begin  
    Ap1 := Ap^.Dir; Ap^.Dir := Ap1^.Esq; Ap1^.Esq := Ap;  
    Ap1^.BitD := Vertical; Ap^.BitD := Vertical;  
    Ap := Ap1;  
end;
```

```
procedure DE (var Ap: Apontador);  
var Ap1, Ap2: Apontador;  
begin  
    Ap1 := Ap^.Dir; Ap2 := Ap1^.Esq;  
    Ap1^.BitE := Vertical; Ap^.BitD := Vertical;  
    Ap1^.Esq := Ap2^.Dir; Ap2^.Dir := Ap1;  
    Ap^.Dir := Ap2^.Esq; Ap2^.Esq := Ap; Ap := Ap2;  
end;
```

Procedimento para Inserir na Árvore SBB

```

procedure Insere (x: Registro; var Ap: Apontador);
var Fim: boolean; IAp: Inclinacao;
    procedure IInsere (x: Registro; var Ap: Apontador;
        var IAp: Inclinacao; var Fim: boolean);
begin
    if Ap = nil
    then begin
        new(Ap); IAp:=Horizontal; Ap^.Reg:=x; Ap^.BitE:=Vertical;
        Ap^.BitD := Vertical; Ap^.Esq := nil; Ap^.Dir := nil;
        Fim := false;
    end
    else
        if x.Chave < Ap^.Reg.Chave
        then begin
            IInsere(x, Ap^.Esq, Ap^.BitE, Fim);
            if not Fim
            then if Ap^.BitE = Horizontal
                then begin
                    if Ap^.Esq^.BitE = Horizontal
                    then begin EE(Ap); IAp := Horizontal; end
                    else if Ap^.Esq^.BitD = Horizontal
                        then begin ED(Ap); IAp := Horizontal; end;
                end
            else Fim := true;
        end
    end
else


```

Procedimento para Inserir na Árvore SBB

```
if x.Chave > Ap^.Reg.Chave
then begin
  IInsere(x, Ap^.Dir, Ap^.BitD, Fim);
  if not Fim
  then if Ap^.BitD = Horizontal
       then begin
            if Ap^.Dir^.BitD = Horizontal
            then begin DD(Ap); IAp := Horizontal; end
            else if Ap^.Dir^.BitE = Horizontal
                 then begin DE(Ap); IAp := Horizontal; end;
            end
       else Fim := true;
  end
else begin
  writeln('Erro: Chave ja esta na arvore');
  Fim := true;
end;
end;
begin
  IInsere(x, Ap, IAp, Fim);
end;
```

Exemplo

- Inserção de uma seqüência de chaves em uma árvore SBB inicialmente vazia.
 1. Árvore à esquerda é obtida após a inserção das chaves 7, 10, 5.
 2. Árvore do meio é obtida após a inserção das chaves 2, 4 na árvore anterior.
 3. Árvore à direita é obtida após a inserção das chaves 9, 3, 6 na árvore anterior.



```

procedure Inicializa (var Dicionario: TipoDicionario);
begin
  Dicionario := nil;
end;

```

Procedimento Retira

- Retira contém um outro procedimento interno de nome IRetira.
- IRetira usa 3 procedimentos internos: EsqCurto, DirCurto, Antecessor.
 - EsqCurto (DirCurto) é chamado quando um nó folha que é referenciado por um apontador vertical é retirado da subárvore à esquerda (direita) tornando-a menor na altura após a retirada;
 - Quando o nó a ser retirado possui dois descendentes, o procedimento Antecessor localiza o nó antecessor para ser trocado com o nó a ser retirado.

Procedimento para Retirar da Árvore SBB

```

procedure Retira (x: Registro; var Ap: Apontador);
var Fim: boolean;
procedure IRetira (x: Registro; var Ap: Apontador; var Fim: boolean);
var Aux: Apontador;
procedure EsqCurto (var Ap: Apontador; var Fim: boolean);
var Ap1: Apontador;
begin { Folha esquerda retirada => arvore curta na altura esquerda }
  if Ap^.BitE = Horizontal
  then begin Ap^.BitE := Vertical; Fim := true; end
  else if Ap^.BitD = Horizontal
    then begin
      Ap1 := Ap^.Dir; Ap^.Dir := Ap1^.Esq; Ap1^.Esq := Ap; Ap := Ap1;
      if Ap^.Esq^.Dir^.BitE = Horizontal
      then begin DE(Ap^.Esq); Ap^.BitE := Horizontal; end
      else if Ap^.Esq^.Dir^.BitD = Horizontal
        then begin DD(Ap^.Esq); Ap^.BitE := Horizontal; end;
      Fim := true;
    end
  else begin
    Ap^.BitD := Horizontal;
    if Ap^.Dir^.BitE = Horizontal
    then begin DE(Ap); Fim := true; end
    else if Ap^.Dir^.BitD = Horizontal
      then begin DD(Ap); Fim := true; end;
    end;
end; { EsqCurto }

```

Procedimento para Retirar da Árvore SBB – DirCurto

```

procedure DirCurto (var Ap: Apontador; var Fim: boolean);
var Ap1: Apontador;
begin { Folha direita retirada => arvore curta na altura direita }
  if Ap^.BitD = Horizontal
  then begin Ap^.BitD := Vertical; Fim := true; end
  else if Ap^.BitE = Horizontal
    then begin
      Ap1 := Ap^.Esq; Ap^.Esq := Ap1^.Dir; Ap1^.Dir := Ap; Ap := Ap1;
      if Ap^.Dir^.Esq^.BitD = Horizontal
      then begin ED(Ap^.Dir); Ap^.BitD := Horizontal; end
      else if Ap^.Dir^.Esq^.BitE = Horizontal
        then begin EE(Ap^.Dir); Ap^.BitD := Horizontal; end;
      Fim := true;
    end
  else begin
    Ap^.BitE := Horizontal;
    if Ap^.Esq^.BitD = Horizontal
    then begin ED(Ap); Fim := true; end
    else if Ap^.Esq^.BitE = Horizontal
      then begin EE(Ap); Fim := true; end;
    end;
end; { DirCurto }

```

Procedimento para Retirar da Árvore SBB – Antecessor

```
procedure Antecessor (q: Apontador; var r: Apontador;  
                    var Fim: boolean);  
begin  
  if r^.Dir <> nil  
  then begin  
    Antecessor(q, r^.Dir, Fim);  
    if not Fim then DirCurto(r, Fim);  
  end  
  else begin  
    q^.Reg := r^.Reg; q := r;  
    r := r^.Esq; dispose(q);  
    if r <> nil then Fim := true;  
  end;  
end; { Antecessor }
```

Procedimento para Retirar da Árvore SBB


```

begin { IRetira }
  if Ap = nil
  then begin writeln( 'Chave nao esta na arvore' ); Fim := true; end
  else if x.Chave < Ap^.Reg.Chave
    then begin
      IRetira(x, Ap^.Esq, Fim);
      if not Fim then EsqCurto(Ap, Fim);
    end
  else if x.Chave > Ap^.Reg.Chave
    then begin
      IRetira(x, Ap^.Dir, Fim);
      if not Fim then DirCurto(Ap, Fim);
    end
  else begin { Encontrou chave }
    Fim := false; Aux := Ap;
    if Aux^.Dir = nil
    then begin
      Ap := Aux^.Esq; dispose(Aux);
      if Ap <> nil then Fim := true;
    end
    else if Aux^.Esq = nil
    then begin
      Ap := Aux^.Dir; dispose(Aux);
      if Ap <> nil then Fim := true;
    end
    else begin
      Antecessor(Aux, Aux^.Esq, Fim);
      if not Fim then EsqCurto(Ap, Fim);
    end;
  end;
end;
begin { Retira }
  IRetira(x, Ap, Fim)
end;

```


Exemplo

- Dada a **Árvore**:





- Resultado obtido quando se retira uma seqüência de chaves da árvore SBB mais à direita acima:

- A árvore à esquerda é obtida após a retirada da chave 7 da árvore à direita acima.
- A árvore do meio é obtida após a retirada da chave 5 da árvore anterior.
- A árvore à direita é obtida após a retirada da chave 9 da árvore anterior.





Exemplo: Retirada de Nós de SBB





Exemplo: Retirada de Nós de SBB

Caso 3:



Se nodo 8 tem filho:



Análise

- Nas árvores SBB é necessário distinguir dois tipos de **alturas**:
 1. Altura vertical $h \rightarrow$ necessária para manter a altura uniforme e obtida através da contagem do número de apontadores verticais em qualquer caminho entre a raiz e um nó externo.
 2. Altura $k \rightarrow$ representa o número máximo de comparações de chaves obtida através da contagem do número total de apontadores no maior caminho entre a raiz e um nó externo.
- A altura k é maior que a altura h sempre que existirem apontadores horizontais na árvore.
- Para uma árvore SBB com n nós internos, temos que

$$h \leq k \leq 2h.$$

Análise

- De fato Bayer (1972) mostrou que

$$\log(n + 1) \leq k \leq 2 \log(n + 2) - 2.$$

- Custo para manter a propriedade SBB \Rightarrow
Custo para percorrer o caminho de pesquisa para encontrar a chave, seja para inserí-la ou para retirá-la.

- **Logo:** O custo é $O(\log n)$.

- Número de comparações em uma pesquisa com sucesso na árvore SBB é

$$\text{melhor caso : } C(n) = O(1)$$

$$\text{pior caso : } C(n) = O(\log n)$$

$$\text{caso médio : } C(n) = O(\log n)$$

- **Observe:** Na prática o caso médio para C_n é apenas cerca de 2% pior que o C_n para uma árvore completamente balanceada, conforme mostrado em Ziviani e Tompa (1982).

Pesquisa Digital

- Pesquisa digital é baseada na representação das chaves como uma seqüência de caracteres ou de dígitos.
- Os métodos de pesquisa digital são particularmente vantajosos quando as chaves são grandes e de **tamanho variável**.
- Um aspecto interessante quanto aos métodos de pesquisa digital é a possibilidade de localizar todas as ocorrências de uma determinada cadeia em um texto, com tempo de resposta logarítmico em relação ao tamanho do texto.
 - **Trie**
 - **Patrícia**

Trie

- Uma trie é uma árvore M -ária cujos nós são vetores de M componentes com campos correspondentes aos dígitos ou caracteres que formam as chaves.
- Cada nó no nível i representa o conjunto de todas as chaves que começam com a mesma seqüência de i dígitos ou caracteres.
- Este nó especifica uma ramificação com M caminhos dependendo do $(i + 1)$ -ésimo dígito ou caractere de uma chave.
- **Considerando as chaves como seqüência de bits (isto é, $M = 2$), o algoritmo de pesquisa digital é semelhante ao de pesquisa em árvore, exceto que, em vez de se caminhar na árvore de acordo com o resultado de comparação entre chaves, caminha-se de acordo com os bits de chave.**

Exemplo

- Dada as chaves de 6 bits:


B = 010010

C = 010011


H = 011000

J = 100001

M = 101000




Inserção das Chaves W e K na Trie Binária



Faz-se uma pesquisa na árvore com a chave a ser inserida. Se o nó externo em que a pesquisa terminar for vazio, cria-se um novo nó externo nesse ponto contendo a nova chave, exemplo: a inserção da chave $W = 110110$.

Se o nó externo contiver uma chave cria-se um ou mais nós internos cujos descendentes conterão a chave já existente e a nova chave. exemplo: inserção da chave $K = 100010$.



Considerações Importantes sobre as Tries

- O formato das tries, diferentemente das árvores binárias comuns, não depende da ordem em que as chaves são inseridas e sim da estrutura das chaves através da distribuição de seus bits.
- **Desvantagem:**
 - Uma grande desvantagem das tries é a formação de caminhos de uma só direção para chaves com um grande número de bits em comum.
 - **Exemplo:** Se duas chaves diferirem somente no último bit, elas formarão um caminho cujo comprimento é igual ao tamanho delas, não importando quantas chaves existem na árvore.
 - Caminho gerado pelas chaves B e C.

Patricia - Practical Algorithm To Retrieve Information Coded In Alphanumeric

- Criado por Morrison D. R. 1968 para aplicação em recuperação de informação em arquivos de grande porte.
- Knuth D. E. 1973 → novo tratamento algoritmo.
- Reapresentou-o de forma mais clara como um caso particular de pesquisa digital, essencialmente, um caso de árvore trie binária.
- Sedgewick R. 1988 apresentou novos algoritmos de pesquisa e de inserção baseados nos algoritmos propostos por Knuth.
- Gonnet, G.H e Baeza-Yates R. 1991 propuzeram também outros algoritmos.

Mais sobre Patricia

- O algoritmo para construção da árvore Patricia é baseado no método de pesquisa digital, mas sem apresentar o inconveniente citado para o caso das tries.
- O problema de caminhos de uma só direção é eliminado por meio de uma solução simples e elegante: cada nó interno da árvore contém o índice do bit a ser testado para decidir qual ramo tomar.
- **Exemplo:** dada as chaves de 6 bits:


B = 010010

C = 010011


H = 011000

J = 100001


Q = 101000



Inserção da Chave K




- Para inserir a chave $K = 100010$ na árvore acima, a pesquisa inicia pela raiz e termina quando se chega ao nó externo contendo J.
- Os índices dos bits nas chaves estão ordenados da esquerda para a direita. Bit de índice 1 de K é 1 → a subárvore direita Bit de índice 3 → subárvore esquerda que neste caso é um **nó externo**.
- Chaves J e K mantêm o padrão de bits $1x0xxx$, assim como qualquer outra chave que seguir este caminho de pesquisa.
- Novo nó interno repõe o nó J, e este com nó K serão os nós externos descendentes.
- O índice do novo nó interno é dado pelo 1º bit diferente das 2 chaves em questão, que é o bit de índice 5. Para determinar qual será o descendente esquerdo e o direito, verifique o valor do bit 5 de ambas as chaves.



Inserção da Chave W

- A inserção da chave $W = 110110$ ilustra um outro aspecto.
- Os bits das chaves K e W são comparados a partir do primeiro para determinar em qual índice eles diferem, sendo, neste caso, os de índice 2.
- **Portanto:** o ponto de inserção agora será no caminho de pesquisa entre os nós internos de índice 1 e 3.
- Cria-se aí um novo nó interno de índice 2, cujo descendente direito é um nó externo contendo W e cujo descendente esquerdo é a subárvore de raiz de índice 3.



Estrutura de Dados

```
const D = 8; { depende de ChaveTipo }  
type ChaveTipo = char; { a definir, dependendo da aplicacao }  
  IndexAmp = 0..D;  
  Dib      = 0..1;  
  NoTipo   = (Interno, Externo);  
  Arvore   = ^PatNo;  
  PatNo    = record  
    case nt: NoTipo of  
      Interno: (Index: IndexAmp; Esq, Dir: Arvore);  
      Externo: (Chave: ChaveTipo);  
    end;
```

Funções Auxiliares

```
function Bit (i: IndexAmp; k: ChaveTipo): Dib;
{ Retorna o i-ésimo bit da chave k a partir da esquerda }
var c, j: integer;
begin
    if i = 0
    then Bit := 0
    else begin
        c := ord (k);
        for j := 1 to D – i do c := c div 2;
        Bit := c mod 2;
    end;
end;

function EExterno (p: Arvore): boolean;
{ Verifica se p^ e no externo }
begin
    EExterno := p^.nt = Externo;
end;
```

Procedimentos para criar os nós

Procedimento para criar nó interno:

```
function CriaNoInt (i: integer; var Esq, Dir: Arvore): Arvore;  
var p: Arvore;  
begin  
    new (p, Interno);  
    p^.nt := Interno;  
    p^.Esq := Esq; p^.Dir := Dir;  
    p^.Index := i; CriaNoInt := p;  
end;
```

Procedimento para criar nó externo:

```
function CriaNoExt (k: ChaveTipo): Arvore;  
var p: Arvore;  
begin  
    new (p, Externo);  
    p^.nt := Externo;  
    p^.Chave := k;  
    CriaNoExt := p;  
end;
```

Algoritmo de pesquisa

```
procedure Pesquisa (k: ChaveTipo; t: Arvore);  
begin  
  if EExterno (t)  
  then if k = t^.Chave  
    then writeln ( 'Elemento encontrado')  
    else writeln ( 'Elemento nao encontrado')  
  else if Bit (t^.Index, k) = 0  
    then Pesquisa (k, t^.Esq)  
    else Pesquisa (k, t^.Dir);  
end;
```

Descrição Informal do Algoritmo de Inserção

- Cada chave k é inserida de acordo com os passos abaixo, partindo da raiz:
 1. Se a subárvore corrente for vazia, então é criado um nó externo contendo a chave k (isto ocorre somente na inserção da primeira chave) e o algoritmo termina.
 2. Se a subárvore corrente for simplesmente um nó externo, os bits da chave k são comparados, a partir do bit de índice imediatamente após o último índice da seqüência de índices consecutivos do caminho de pesquisa, com os bits correspondentes da chave k' deste nó externo até encontrar um índice i cujos bits difiram. A comparação dos bits a partir do último índice consecutivo melhora consideravelmente o desempenho do algoritmo. Se todos forem iguais, a chave já se encontra na árvore e o algoritmo termina; senão, vai-se para o Passo 4.

Descrição Informal do Algoritmo de Inserção

- Continuação:
 3. Se a raiz da subárvore corrente for um nó interno, vai-se para a subárvore indicada pelo bit da chave k de índice dado pelo nó corrente, de forma recursiva.
 4. Depois são criados um nó interno e um nó externo: o primeiro contendo o índice i e o segundo, a chave k . A seguir, o nó interno é ligado ao externo pelo apontador de subárvore esquerda ou direita, dependendo se o bit de índice i da chave k seja 0 ou 1, respectivamente.
 5. O caminho de inserção é percorrido novamente de baixo para cima, subindo com o par de nós criados no Passo 4 até chegar a um nó interno cujo índice seja menor que o índice i determinado no Passo 2. Este é o ponto de inserção e o par de nós é inserido.

Algoritmo de inserção

```
function Insere (k: ChaveTipo; var t: Arvore): Arvore;
var p: Arvore; i: integer;

function InsereEntre(k: ChaveTipo; var t: Arvore;
                    i: integer): Arvore;

var p: Arvore;
begin
  if EExterno (t) or (i < t^.Index)
  then begin { cria um novo no externo }
    p := CriNoExt (k);
    if Bit (i, k) = 1
    then InsereEntre := CriNoInt (i, t, p)
    else InsereEntre := CriNoInt (i, p, t);
    end
  else begin
    if Bit (t^.Index, k) = 1
    then t^.Dir := InsereEntre (k, t^.Dir, i)
    else t^.Esq := InsereEntre (k, t^.Esq, i);
    InsereEntre := t;
    end;
end;
```

Algoritmo de inserção

```
begin {Insere}
  if t = nil
  then Insere := CriaNoExt (k)
  else begin
    p := t;
    while not EExterno (p) do
      begin
        if Bit(p^.Index, k) = 1
        then p := p^.Dir
        else p := p^.Esq;
        end;
        i := 1; { acha o primeiro bit diferente }
        while (i <= D) and (Bit(i, k) = Bit(i, p^.Chave)) do
          i := i + 1;
        if i > D
        then begin
          writeln ('Erro: chave ja esta na arvore');
          Insere := t;
          end
        else Insere := InsereEntre (k, t, i);
      end;
end; {Insere}
```

Transformação de Chave (*Hashing*)

- Os registros armazenados em uma tabela são diretamente endereçados a partir de uma transformação aritmética sobre a chave de pesquisa.
- *Hash* significa:
 1. Fazer picadinho de carne e vegetais para cozinhar.
 2. Fazer uma bagunça. (Webster's New World Dictionary)

Transformação de Chave (*Hashing*)

- Um método de pesquisa com o uso da transformação de chave é constituído de duas etapas principais:
 1. Computar o valor da **função de transformação**, a qual transforma a chave de pesquisa em um endereço da tabela.
 2. Considerando que duas ou mais chaves podem ser transformadas em um mesmo endereço de tabela, é necessário existir um método para lidar com **colisões**.
- Qualquer que seja a função de transformação, algumas **colisões** irão ocorrer fatalmente, e tais colisões têm de ser resolvidas de alguma forma.
- Mesmo que se obtenha uma função de transformação que distribua os registros de forma uniforme entre as entradas da tabela, existe uma alta probabilidade de haver colisões.

Transformação de Chave (*Hashing*)

- O **paradoxo do aniversário** (Feller, 1968, p. 33), diz que em um grupo de 23 ou mais pessoas, juntas ao acaso, existe uma chance maior do que 50% de que 2 pessoas comemorem aniversário no mesmo dia.
- Assim, se for utilizada uma função de transformação uniforme que enderece 23 chaves randômicas em uma tabela de tamanho 365, a probabilidade de que haja **colisões** é maior do que 50%.
- A probabilidade p de se inserir N itens consecutivos sem colisão em uma tabela de tamanho M é:

$$p = \frac{M-1}{M} \times \frac{M-2}{M} \times \dots \times \frac{M-N+1}{M} = \prod_{i=1}^N \frac{M-i+1}{M} = \frac{M!}{(M-N)!M^N}$$

Transformação de Chave (*Hashing*)

- Alguns valores de p para diferentes valores de N , onde $M = 365$.

N	p
10	0,883
22	0,524
23	0,493
30	0,303

- Para N pequeno a probabilidade p pode ser aproximada por $p \approx \frac{N(N-1)}{730}$. Por exemplo, para $N = 10$ então $p \approx 87,7\%$.

Funções de Transformação

- Uma função de transformação deve mapear chaves em inteiros dentro do intervalo $[0..M - 1]$, onde M é o tamanho da tabela.
- **A função de transformação ideal é aquela que:**
 1. Seja simples de ser computada.
 2. Para cada chave de entrada, qualquer uma das saídas possíveis é igualmente provável de ocorrer.

Método mais Usado

- Usa o resto da divisão por M .

$$h(K) = K \bmod M$$

onde K é um inteiro correspondente à chave.

- **Cuidado** na escolha do valor de M . M deve ser um **número primo**, mas não qualquer primo: devem ser evitados os números primos obtidos a partir de

$$b^i \pm j$$

onde b é a base do conjunto de caracteres (geralmente $b = 64$ para BCD, 128 para ASCII, 256 para EBCDIC, ou 100 para alguns códigos decimais), e i e j são pequenos inteiros.

Transformação de Chaves Não Numéricas

- As chaves não numéricas devem ser transformadas em números:

$$K = \sum_{i=1}^n \text{Chave}[i] \times p[i],$$

- n é o número de caracteres da chave.
- $\text{Chave}[i]$ corresponde à representação ASCII do i -ésimo caractere da chave.
- $p[i]$ é um inteiro de um conjunto de pesos gerados randomicamente para $1 \leq i \leq n$.
- Vantagem de se usar pesos: Dois conjuntos diferentes de pesos $p_1[i]$ e $p_2[i]$, $1 \leq i \leq n$, leva a duas funções de transformação $h_1(K)$ e $h_2(K)$ diferentes.

Transformação de Chaves Não Numéricas

- Programa que gera um peso para cada caractere de uma chave constituída de n caracteres:

```
type TipoPesos = array [1..n] of integer;
```

```
procedure GeraPesos(var p: TipoPesos);
```

```
var i: integer;
```

```
begin
```

```
    randomize;
```

```
    for i := 1 to n do
```

```
        p[i] := trunc (1000000 * random + 1);
```

```
end;
```

- **Implementação da função de transformação:**

```
type TipoChave = packed array [1..n] of char;
```

```
    Indice = 0 .. M - 1;
```

```
function h(Chave: TipoChave; p: TipoPesos): Indice;
```

```
var i, Soma: integer;
```

```
begin
```

```
    Soma := 0;
```


```
    for i := 1 to n do Soma := Soma + ord(Chave[i]) * p[i];
```

```
    h := Soma mod M;
```

```
end;
```

Listas Encadeadas

- Uma das formas de resolver as **colisões** é simplesmente construir uma lista linear encadeada para cada endereço da tabela. Assim, todas as chaves com mesmo endereço são encadeadas em uma lista linear.
- **Exemplo:** Se a i -ésima letra do alfabeto é representada pelo número i e a função de transformação $h(Chave) = Chave \bmod M$ é utilizada para $M = 7$, o resultado da inserção das chaves $P E S Q U I S A$ na tabela é o seguinte:
- Por exemplo, $h(A) = h(1) = 1$, $h(E) = h(5) = 5$, $h(S) = h(19) = 5$, e assim por diante.



Estrutura do Dicionário para Listas Encadeadas

type

TipoChave = **array** [1..n] **of** **char**;

Indice = 0..M-1;

TipoItem = **record**

 Chave: TipoChave

 { *outros componentes* }

end;

Apontador = ^ Celula;

Celula = **record**

 Item: TipoItem;

 Prox: Apontador

end;

TipoLista = **record**

 Primeiro: Apontador;

 Ultimo : Apontador

end;

TipoDicionario = **array** [Indice] **of** TipoLista;

Operações do Dicionário Usando Listas Encadeadas

```

procedure Inicializa (var T: TipoDicionario);
var i: integer;
begin
    for i := 0 to M - 1 do FLVazia(T[i])
end;

function Pesquisa (Ch: TipoChave; var p: TipoPesos;
                   var T: TipoDicionario): Apontador;
    {— Obs.: Apontador de retorno aponta para o item
     anterior da lista —}
var i: Indice; Ap: Apontador;
begin
    i := h(Ch, p);
    if Vazia(T[i])
    then Pesquisa := nil { Pesquisa sem sucesso }
    else begin
        Ap := T[i].Primeiro;
        while (Ap^.Prox^.Prox <> nil) and
            (Ch <> Ap^.Prox^.Item.Chave) do
            Ap := Ap^.Prox;
        if Ch = Ap^.Prox^.Item.Chave
        then Pesquisa := Ap
        else Pesquisa := nil { Pesquisa sem sucesso }
    end
end;

```

Operações do Dicionário Usando Listas Encadeadas

```
procedure Insere (x: TipoItem; var p: TipoPesos;  
                 var T: TipoDicionario);
```

```
begin
```

```
    if Pesquisa(x.Chave, p, T) = nil
```

```
    then Ins(x, T[h(x.Chave, p)])
```

```
    else writeln(' Registro ja esta presente')
```

```
end;
```

```
procedure Retira (x: TipoItem; var p: TipoPesos;  
                 var T: TipoDicionario);
```

```
var Ap: Apontador;
```

```
begin
```

```
    Ap := Pesquisa(x.Chave, p, T);
```

```
    if Ap = nil
```

```
    then writeln(' Registro nao esta presente')
```

```
    else Ret(Ap, T[h(x.Chave, p)], x)
```

```
end;
```

Análise

- Assumindo que qualquer item do conjunto tem igual probabilidade de ser endereçado para qualquer entrada de T , então o comprimento esperado de cada lista encadeada é N/M , onde N representa o número de registros na tabela e M o tamanho da tabela.
- **Logo:** as operações Pesquisa, Insere e Retira custam $O(1 + N/M)$ operações em média, onde a constante 1 representa o tempo para encontrar a entrada na tabela e N/M o tempo para percorrer a lista. Para valores de M próximos de N , o tempo se torna constante, isto é, independente de N .

Endereçamento Aberto

- Quando o número de registros a serem armazenados na tabela puder ser previamente estimado, então não haverá necessidade de usar apontadores para armazenar os registros.
- Existem vários métodos para armazenar N registros em uma tabela de tamanho $M > N$, os quais utilizam os lugares vazios na própria tabela para resolver as **colisões**. (Knuth, 1973, p.518)
- No **Endereçamento aberto** todas as chaves são armazenadas na própria tabela, sem o uso de apontadores explícitos.
- Existem várias propostas para a escolha de localizações alternativas. A mais simples é chamada de **hashing linear**, onde a posição h_j na tabela é dada por:

$$h_j = (h(x) + j) \bmod M, \quad \text{para } 1 \leq j \leq M - 1.$$

Exemplo

- Se a i -ésima letra do alfabeto é representada pelo número i e a função de transformação $h(Chave) = Chave \bmod M$ é utilizada para $M = 7$,
- então o resultado da inserção das chaves $L U N E S$ na tabela, usando *hashing linear* para resolver colisões é mostrado abaixo.
- Por exemplo, $h(L) = h(12) = 5$,
 $h(U) = h(21) = 0$, $h(N) = h(14) = 0$,
 $h(E) = h(5) = 5$, e $h(S) = h(19) = 5$.

T

0	U
1	N
2	S
3	
4	
5	L
6	E

Estrutura do Dicionário Usando Endereçamento Aberto

```
const Vazio    = '!!!!!!!!!!!';
      Retirado = '*****';
      M = 7;
      n = 10; { Tamanho da chave }

type Apontador = integer;
      TipoChave = packed array [1..n] of char;
      TipoPesos = array [1..n] of integer;
      Tipoltem  = record
                  Chave: TipoChave
                  { outros componentes }
                  end;
      Indice    = 0 .. M - 1;
      TipoDicionario = array [Indice] of Tipoltem;
```

Operações do Dicionário Usando Endereçamento Aberto

```
procedure Inicializa (var T: TipoDicionario);
var i: integer;
begin
    for i := 0 to M - 1 do T[i].Chave := Vazio
end;

function Pesquisa (Ch: TipoChave; var p: TipoPesos;
                 var T: TipoDicionario): Apontador;
var i: integer;
    Inicial: integer;
begin
    Inicial := h(Ch, p);
    i := 0;
    while (T[(Inicial + i) mod M].Chave <> Vazio) and
          (T[(Inicial + i) mod M].Chave <> Ch) and
          (i < M) do i := i + 1;
    if T[(Inicial + i) mod M].Chave = Ch
    then Pesquisa := (Inicial + i) mod M
    else Pesquisa := M { Pesquisa sem sucesso }
end;
```

Operações do Dicionário Usando Endereçamento Aberto

```
procedure Insere (x: TipoItem; var p: TipoPesos;
                 var T: TipoDicionario);
var i: integer;
    Inicial: integer;
begin
    if Pesquisa(x.Chave, p, T) < M
    then writeln('Elemento ja esta presente')
    else begin
        Inicial := h(x.Chave, p);
        i := 0;
        while ((T[(Inicial + i) mod M].Chave  $\neq$  Vazio) and
              (T[(Inicial + i) mod M].Chave  $\neq$  Retirado)) and
              (i < M) do i := i + 1;
        if i < M
        then T[(Inicial + i) mod M] := x
        else writeln(' Tabela cheia ')
    end;
end;
```

Operações do Dicionário Usando Endereçamento Aberto

```
procedure Retira(Ch:TipoChave; var p:TipoPesos;
                var T:TipoDicionario);
var i: Indice;
begin
  i := Pesquisa(Ch, p, T);
  if i < M
  then T[i].Chave := Retirado
  else writeln('Registro nao esta presente')
end;
```

Análise

- Seja $\alpha = N/M$ o fator de carga da tabela. Conforme demonstrado por Knuth (1973), o custo de uma pesquisa com sucesso é

$$C(n) = \frac{1}{2} \left(1 + \frac{1}{1 - \alpha} \right).$$

- O *hashing linear* sofre de um mal chamado **agrupamento(*clustering*)** (Knuth, 1973, pp.520–521).
- Este fenômeno ocorre na medida em que a tabela começa a ficar cheia, pois a inserção de uma nova chave tende a ocupar uma posição na tabela que esteja contígua a outras posições já ocupadas, o que deteriora o tempo necessário para novas pesquisas.
- Entretanto, apesar do *hashing linear* ser um método relativamente pobre para resolver colisões os resultados apresentados são bons.
- O melhor caso, assim como o caso médio, é $O(1)$.

Vantagens e Desvantagens de Transformação da Chave

Vantagens:

- Alta eficiência no custo de pesquisa, que é $O(1)$ para o caso médio.
- Simplicidade de implementação.

Desvantagens:

- Custo para recuperar os registros na ordem lexicográfica das chaves é alto, sendo necessário ordenar o arquivo.
- Pior caso é $O(N)$.

Hashing Perfeito

- Se $h(x_i) = h(x_j)$ se e somente se $i = j$, então não há colisões, e a função de transformação é chamada de **função de transformação perfeita** ou função *hashing* perfeita(hp).
- Se o número de chaves N e o tamanho da tabela M são iguais ($\alpha = N/M = 1$), então temos uma **função de transformação perfeita mínima**.
- Se $x_i \leq x_j$ e $hp(x_i) \leq hp(x_j)$, então a ordem lexicográfica é preservada. Nesse caso, temos uma **função de transformação perfeita mínima com ordem preservada**.

Vantagens e Desvantagens de Uma Função de Transformação Perfeita

- Não há necessidade de armazenar a chave, pois o registro é localizado sempre a partir do resultado da função de transformação.
- Uma função de transformação perfeita é específica para um conjunto de chaves conhecido.
- A desvantagem no caso é o espaço ocupado para descrever a função de transformação hp .
- Entretanto, é possível obter um método com $M \approx 1,25N$, para valores grandes de N .

Algoritmo de Czech, Havas e Majewski

- Czech, Havas e Majewski (1992, 1997) propõem um método elegante baseado em **grafos randômicos** para obter uma função de transformação perfeita com ordem preservada.
- A função de transformação é do tipo:

$$hp(x) = (g(h_1(x)) + g(h_2(x))) \bmod N,$$

na qual $h_1(x)$ e $h_2(x)$ são duas funções não perfeitas, x é a chave de busca, e g um arranjo especial que mapeia números no intervalo $0 \dots M - 1$ para o intervalo $0 \dots N - 1$.

Problema Resolvido Pelo Algoritmo

- Dado um grafo não direcionado $G = (V, A)$, onde $|V| = M$ e $|A| = N$, encontre uma função $g : V \rightarrow [0, N - 1]$, definida como $hp(a = (u, v) \in A) = (g(u) + g(v)) \bmod N$.
- Em outras palavras, estamos procurando uma atribuição de valores aos vértices de G tal que a soma dos valores associados aos vértices de cada aresta tomado módulo N é um número único no intervalo $[0, N - 1]$.
- A questão principal é como obter uma função g adequada. A abordagem mostrada a seguir é baseada em grafos e hipergrafos randômicos.


Exemplo

- **Chaves:** 12 meses do ano abreviados para os três primeiros caracteres.
- **Objetivo:** obter uma função de transformação perfeita h_p de tal forma que o i -ésimo mês é mantido na $(i - 1)$ -ésima posição da tabela *hash*:

Chave x	$h_1(x)$	$h_2(x)$	$h_p(x)$
jan	10	11	0
fev	1	2	1
mar	8	9	2
abr	1	3	3
mai	0	5	4
jun	10	9	5
jul	0	3	6
ago	5	6	7
set	4	1	8
out	0	1	9
nov	3	2	10
dez	4	7	11

Grafo Randômico gerado

- O problema de obter a função g é equivalente a encontrar um grafo não direcionado contendo M vértices e N arestas.




- Os vértices são rotulados com valores no intervalo $0 \dots M - 1$
- As arestas definidas por $(h_1(x), h_2(x))$ para cada uma das N chaves x .
- Cada chave corresponde a uma aresta que é rotulada com o valor desejado para a função h_p perfeita.
- Os valores das duas funções $h_1(x)$ e $h_2(x)$ definem os vértices sobre os quais a aresta é incidente.

Obtenção da Função g a Partir do Grafo

- **Passo importante:** conseguir um arranjo g de vértices para inteiros no intervalo $0 \dots N - 1$ tal que, para cada aresta $(h_1(x), h_2(x))$, o valor de $hp(x) = g(h_1(x)) + g(h_2(x)) \pmod N$ seja igual ao rótulo da aresta.
- **Algoritmo:**
 1. Qualquer vértice não processado é escolhido e feito $g[v] = 0$.
 2. As arestas que saem do vértice v são seguidas e o valor $g(u)$ do vértice u destino é rotulado com o valor da diferença entre o valor da aresta (v, u) e $g(v)$, tomado $\pmod N$.
 3. Procura-se o próximo componente conectado ainda não visitado e os mesmos passos descritos acima são repetidos.

Seguindo o Algoritmo para Obter g no Exemplo dos 12 Meses do Ano



	Chave x	$h_1(x)$	$h_2(x)$	$hp(x)$		$v :$	$g(v)$
(a)	jan	10	11	0	(b)	0	0
	fev	1	2	1		1	9
	mar	8	9	2		2	4
	abr	1	3	3		3	6
	mai	0	5	4		4	11
	jun	10	9	5		5	4
	jul	0	3	6		6	3
	ago	5	6	7		7	0
	set	4	1	8		8	0
	out	0	1	9		9	2
	nov	3	2	10		10	3
	dez	4	7	11		11	9

Problema

- **Quando o grafo contém ciclos:** o mapeamento a ser realizado pode rotular de novo um vértice já processado e que tenha recebido outro rótulo com valor diferente.
- Por exemplo, se a aresta $(5, 6)$, que é a aresta de rótulo 7, tivesse sido sorteada para a aresta $(8, 11)$, o algoritmo tentaria atribuir dois valores distintos para o valor de $g[11]$.
- Para enxergar isso, vimos que se $g[8] = 0$, então $g[11]$ deveria ser igual a 7, e não igual ao valor 9 obtido acima.
- Um grafo que permite a atribuição de dois valores de g para um mesmo vértice, não é válido.
- Grafos acíclicos não possuem este problema.
- Um caminho seguro para se ter sucesso é obter antes um grafo acíclico e depois realizar a atribuição de valores para o arranjo g .
Czech, Havas e Majewski (1992).

Primeiro Refinamento do Procedimento para Atribuir Valores ao Arranjo g

```
Procedure Atribuig (G: TipoGrafo; g: Tipog;  
                    GrafoRotulavel: boolean);  
  
  procedure RotuleDe (v, c);  
  begin  
    if g[v] <> Indefinido  
  then if g[v] <> c then GrafoRotulavel := false  
  else begin  
    g[v] := c;  
    for u ∈ ListaAdjacentes(v) do  
      RotuleDe (u, (Aresta(v,u) – g[v]) mod N)  
    end;  
  end;  
  
  begin  
    GrafoRotulavel := true;  
    for v := 0 to M–1 do g[v] := Indefinido;  
    for v := 0 to M–1 do  
      if g[v] Indefinido then RotuleDe (v, 0);  
  end;
```

Algoritmo para Obter a Função de Transformação Perfeita

Program ObtemHashingPerfeito;

begin

Ler conjunto de N chaves;

Escolha um valor para M ;

repeat

 Gera os pesos $p_1[i]$ e $p_2[i]$ para $1 \leq i \leq MaxTamChave$;

 Gera o grafo $G = (V, A)$;

 Atribui $(G, g, GrafoRotulavel)$;

until GrafoRotulavel;

Retorna $p_1[i]$, $p_2[i]$ e g ;

end.

Estruturas de dados

const

```

MaxNumVertices = 100000; { Numero maximo de vertices }
MaxNumArestas  = 100000; { Numero maximo de arestas  }
MaxTamChave    = 6;
MaxNumChaves   = 100000; { Numero maximo de chaves lidas }
MaxTam         = MaxNumVertices+2*MaxNumArestas;
Indefinido     = -1;

```

type

```

TipoValorVertice = 0..MaxNumVertices;
TipoValorAresta  = 0..MaxNumArestas;
TipoPesos        = array [1..MaxTamChave] of integer;
Tipog            = array[0..MaxNumVertices] of integer;
TipoChave       = packed array[1..MaxTamChave] of char;
TipoConjChaves  = array[0..MaxNumChaves] of TipoChave;
Indice          = TipoValorVertice;
TipoTam        = 0..MaxTam;
TipoGrafo = record
    Cab          : array[TipoTam] of TipoTam;
    Prox         : array[TipoTam] of TipoTam;
    Aresta      : array[TipoTam] of TipoTam;
    ProxDisponivel: TipoTam;
    NumVertices : 0..MaxNumvertices;
    NumArestas  : 0..MaxNumArestas;
end;
Apontador = TipoTam;

```

Gera um Grafo sem Arestas Repetidas e sem *Self-Loops*

```
procedure GeraGrafo(var ConjChaves      : TipoConjChaves;
                   var p1, p2          : TipoPesos;
                   var NgrafosGerados: integer;
                   var Grafo          : TipoGrafo);
{-- Entram aqui GeraPesos, funcao h--}
{-- Operadores do tipo abstrato de dados Grafo--}
var i, j                : integer;
    Vertice1, Vertice2: TipoValorVertice;
    GrafoValido        : boolean;
begin
    repeat
        GrafoValido := true;
        FGVazio(Grafo);
        GeraPesos(p1);
        GeraPesos(p2);
        for i := 0 to Grafo.NumArestas-1 do
            begin
                Vertice1 := h(ConjChaves[i], p1);
                Vertice2 := h(ConjChaves[i], p2);
```

Gera um Grafo sem Arestas Repetidas e sem *Self-Loops*

```
    if (Vertice1 = Vertice2) or
        ExisteAresta(Vertice1 , Vertice2 , Grafo)
    then begin
        GrafoValido := false;
        LiberaGrafo(Grafo);
        break;
    end
    else begin
        Aresta := i;
        InsereAresta(Vertice1 , Vertice2 , Aresta ,
                    Grafo);
        InsereAresta(Vertice2 , Vertice1 , Aresta ,
                    Grafo);
    end
end;
NGrafosGerados := NGrafosGerados + 1;
until GrafoValido;
end;
```

Rotula Grafo e Atribui Valores para O Arranjo g

```

Procedure Atribuig (var Grafo           : TipoGrafo;
                    var p1, p2         : TipoPesos;
                    var g             : Tipog;
                    var NGrafosConsiderados: integer;
                    var GrafoRotulavel  : boolean);

var v: TipoValorVertice;
    procedure RotuleDe (v: TipoValorVertice;
                       c: TipoValorAresta;
                       var GrafoRotulavel: boolean);

    var
        Aux           : Apontador;
        u, VerticeAdj: TipoValorVertice;
        Aresta       : TipoValorAresta;
        FimListaAdj  : boolean;

    begin
        if g[v] <> Indefinido
        then begin
            if g[v] <> c
            then begin
                NGrafosConsiderados:=NGrafosConsiderados+1;
                GrafoRotulavel := false;
                LiberaGrafo(Grafo);
            end
        end
    end

```

Rotula Grafo e Atribui Valores para O Arranjo g

```
else begin
    g[v] := c;
    if not ListaAdjVazia(v, Grafo)
    then begin
        Aux := PrimeiroListaAdj(v, Grafo);
        FimListaAdj := false;
        while FimListaAdj = false do
            begin
                ProxAdj(v, VerticeAdj, Aresta, Aux,
                    FimListaAdj);
                u := Aresta - g[v];
                if u < 0 then u := u + N;
                RotuleDe (VerticeAdj, u,
                    GrafoRotulavel);
                if GrafoRotulavel = false
                then break; { Sai do loop }
                end;
            end;
        end;
    end;
end;
```

Rotula Grafo e Atribui Valores para O Arranjo g

```
begin {Atribuig}  
GrafoRotulavel := true;  
for v := 0 to M-1 do g[v] := Indefinido;  
for v := 0 to M-1 do  
    begin  
        if g[v]= Indefinido  
        then RotuleDe (v, 0, GrafoRotulavel);  
        if GrafoRotulavel = false then break;  
    end;  
end; {Atribuig}
```

Programa principal

```
program RotulaGrafoListaPC;  
var  
    M           : TipoValorVertice;  
    N           : TipoValorAresta;  
    Grafo       : TipoGrafo;  
    Aresta     : TipoValorAresta;  
    g           : Tipog;  
    p1, p2     : TipoPesos;  
    i           : integer;  
    NGrafosGerados : integer;  
    NGrafosConsiderados: integer;  
    ConjChaves  : TipoConjChaves;  
    GrafoRotulavel : boolean;  
    ArqEntrada  : text;  
    ArqSaida    : text;  
    NomeArq     : string [100];  
begin { RotulaGrafoListaPC }  
    { Inicializa procedimento Randon para 2^32 valores }  
    randomize;  
    write( 'Nome do arquivo com chaves a serem lidas: ');  
    readln(NomeArq);  
    assign(ArqEntrada, NomeArq);  
    write( 'Nome do arquivo para gravar experimento: ');  
    readln(NomeArq);
```

Programa principal

```
assign(ArqSaida, NomeArq);
reset(ArqEntrada); rewrite(ArqSaida);
NGrafosGerados := 0; NGrafosConsiderados := 0; i := 0;
readln(ArqEntrada, N, M);
{ Numero de entradas da tabela hash }
Grafo.NumArestas := N;
Grafo.NumVertices := M; { Tamanho do arranjo g }
while ( i < N) and (not eof(ArqEntrada)) do
  begin
    readln(ArqEntrada, ConjChaves[ i ]);
    i := i + 1;
  end;
if ( i <> N)
then begin
  writeln( 'Erro: arquivo de entrada possui
            menos que ', N, ' elementos. ');
  exit;
end;
repeat
  GeraGrafo(ConjChaves,p1,p2,NgrafosGerados,Grafo);
  Atribuig (Grafo, p1, p2, g, NGrafosConsiderados,
           GrafoRotulavel);
until GrafoRotulavel;
```

Programa principal

```
writeln(ArqSaida, N, ' (N) ');
writeln(ArqSaida, M, ' (M) ');
for i := 1 to MaxTamChave do write(ArqSaida,p1[i], ' ');
writeln(ArqSaida, ' (p1) ');
for i := 1 to MaxTamChave do write(ArqSaida,p2[i], ' ');
writeln(ArqSaida, ' (p2) ');
for i := 0 to M-1 do write(ArqSaida, g[i], ' ');
writeln(ArqSaida, ' (g) ');
writeln(ArqSaida, 'No. grafos gerados por GeraGrafo:',
        NGrafosGerados);
writeln(ArqSaida, 'No. grafos considerados por
        Atribuig:', NGrafosConsiderados+1);
LiberaGrafo(Grafo);
close(ArqSaida);
close(ArqEntrada);
end. { RotulaGrafoListaPC }
```

Função de Transformação Perfeita

```
function hp(Chave:TipoChave; var p1,p2:TipoPesos;  
            var g:Tipog):Indice;  
begin  
    hp := (g[h(Chave, p1)] + g[h(Chave, p2)]) mod N;  
end;
```

Análise

- **A questão crucial é:** quantas interações são necessárias para obter um grafo $G = (V, A)$ que seja rotulável?
- Para grafos arbitrários, é difícil achar uma solução para esse problema, isso se existir tal solução.
- Entretanto, para **grafos acíclicos**, a função g existe sempre e pode ser obtida facilmente.
- Assim, a resposta a esta questão depende do valor de M que é escolhido no primeiro passo do algoritmo.
- Quanto maior o valor de M , mais esparsos são os grafos e, conseqüentemente, mais provável que eles sejam acíclicos.

Análise

- Segundo Czech, Havas e Majewski (1992), quando $M \leq 2N$ a probabilidade de gerar aleatoriamente um grafo acíclico tende para zero quando N cresce.
- Isto ocorre porque o grafo se torna denso, e o grande número de arestas pode levar à formação de ciclos.
- Por outro lado, quando $M > 2N$, a probabilidade de que um grafo randômico contendo M vértices e N arestas seja acíclico é aproximadamente

$$\sqrt{\frac{M - 2N}{M}},$$

- E o número esperado de grafos gerados até que o primeiro acíclico seja obtido é:

$$\sqrt{\frac{M}{M - 2N}}.$$

Análise

- Para $M = 3N$ o número esperado de iterações é $\sqrt{3}$, \Rightarrow em média, aproximadamente 1,7 grafos serão testados antes que apareça um grafo acíclico.
- Logo, a complexidade de tempo para gerar a função de transformação é proporcional ao número de chaves a serem inseridas na tabela *hash*, desde que $M > 2N$.
- O grande inconveniente de usar $M = 3N$ é o espaço necessário para armazenar o arranjo g .
- Por outro lado, considerar $M < 2N$ pode implicar na necessidade de gerar muitos gráficos randômicos até que um grafo acíclico seja encontrado.

Outra Alternativa

- Não utilizar grafos tradicionais, mas sim **hipergrafos**, ou r -grafos, nos quais cada aresta conecta um número qualquer r de vértices.
- Para tanto, basta usar uma terceira função h_3 para gerar um trigrafo com arestas conectando três vértices, chamado de 3-grafo.
- Em outras palavras, cada aresta é uma tripla do tipo $(h_1(x), h_2(x), h_3(x))$, e a função de transformação é dada por:

$$h(x) = (g(h_1(x)) + g(h_2(x)) + g(h_3(x))) \bmod N.$$

Outra Alternativa

- Nesse caso, o valor de M pode ser próximo a $1,23N$.
- Logo, o uso de trigramas reduz o custo de espaço da função de transformação perfeita, mas aumenta o tempo de acesso ao dicionário.
- Além disso, o processo de rotulação não pode ser feito como descrito.
- Ciclos devem ser detectados previamente, utilizando a seguinte propriedade de r -grafos:

Um r -grafo é **acíclico** se e somente se a remoção repetida de arestas contendo apenas vértices de grau 1 (isto é, vértices sobre os quais incide apenas uma aresta) elimina todas as arestas do grafo.

Experimentos

# Chaves	# Chamadas GeraGrafo	# Chamadas Atribuig	Tempo (s)
10	12	2	0,02
20	189	21	0,13
30	1517	173	1,06
40	21	3	0.04
50	669	110	0,63
60	267	27	0,17
70	6	2	0,02
80	892	117	0,82
90	122	16	0,13
100	808	135	0,89